
Subject: Re: Possible to create 'keyword_used' type function?

Posted by [Chip Helms](#) on Mon, 09 Jun 2014 22:42:16 GMT

[View Forum Message](#) <> [Reply to Message](#)

Here's a cleaner version with some documentation as well.

```
;=====
; PROGRAMMER: Chip Helms
; DATE: 6/9/2014
; DESCRIPTION: Determines if a variable
;   is being used in any way (either having
;   been given a value or set equal to a named
;   variable as often used with keywords).
;   It is equivalent to using:
;   "n_elements(var) ne 0 or arg_present(var)"
;   Note, the result does not have any meaning
;   when called from $MAIN$ as calling the
;   function automatically guarantees 'key' is
;   set to a named variable or has a value.
;   The exception to this is the trivial case
;   of 'keyword_used()', which will return 0.
; PARAMETERS:
;   key := variable to be tested for use
;=====
function keyword_used, key
  ; grab list of variables at scope of parent routine
  varnames = scope_varname(key,level=-2, count=nvar)
  ; determine if keyword is used as per arg_present function
  ; also protect against varnames being undefined for some reason
  argtest = (nvar ne 0) ? total(varnames ne "") ne 0 : 0b
  ; return results of test for keyword presence
  return, n_elements(key) ne 0 or argtest
end
```
